INT 145/88

PROGRAMME ISTART4

Courant d'accrochage d'un gyrotron travaillant en mode TE_{mpq}

André PERRENOUD

Juillet 88

Centre de Recherches en Physique des Plasmas
Association Euratom - Confédération Suisse
Ecole Polytechnique Fédérale de Lausanne

TABLE DES MATIERES

1. BUT DU PROGRAMME ISTART4

Dessiner le courant d'accrochage en fonction du champ magnétique statique B , pour différents modes TE_{mpq}.
A la différence du code ISTART3, on tient compte cette fois de l'indice axial q.

2. MODELE ET EQUATIONS UTILISEES

Le calcul analytique est basé sur les hypothèses suivantes:

Champ électrique dans la cavité cylindrique [1, 2]:

$$\underline{E} = (E_r \hat{r} + E_\phi \hat{\phi})\, e^{i(\omega t + \psi)}$$ avec:

$$\begin{cases} E_r = i \dfrac{m}{k_\perp r} E_0 g(z)\, J_m(k_\perp r)\, e^{-im\phi} \\ E_\phi = E_0 g(z)\, J'_m(k_\perp r)\, e^{-im\phi} \end{cases}$$

A un instant donné, le champ électrique physiquement mesurable est la partie réelle de $\underline{E}$.

$k_\perp = \dfrac{\nu_{mp}}{a}$; ν_{mp} : $p^{\text{ième}}$ zéro non nul de $J'_m(x)$

a = rayon de la cavité

Le champ E a un profil sinusoïdal selon z, l'axe du faisceau d'électrons:

$g(z) = \sin(k_{/\!/} z)$ avec $k_{/\!/} = \dfrac{q\pi}{L}$

L = longueur de la cavité ($\neq$ L du programme ISTART3)

L = 3/2·FWHM de g(z)

La pulsation propre de la cavité est donnée par

$$\left(\frac{\omega}{c}\right)^2 = k^2 = k_\perp^2 + k_{/\!/}^2 \quad \text{d'où} \quad \omega = c\sqrt{\left(\frac{\nu_{mp}}{a}\right)^2 + \left(\frac{q\pi}{L}\right)^2}$$

---

[1] Generalized Nonlinear Harmonic Gyrotron Theory, B.G. Danly and R.J. Temkin, PFC, MIT, April 85.

[2] High Frequency Gyrotrons and Their Application to Tokamak Plasma Heating, K.E. Kreisher and R.J. Temkin, Infrared and MM Waves, Vol. 7, chap. 8, 1983.

Le faisceau électronique est caractérisé par sa tension V_0, son anisotropie $\alpha = \beta_\perp / \beta_\parallel$, son rayon R_e.

$$\gamma_0 = 1 + \frac{eV_0}{mc^2} \quad ; \quad \beta_\parallel = \frac{w}{c} = \sqrt{\frac{1-\gamma_0^{-2}}{1+\alpha^2}} \quad ; \quad \beta_\perp = \frac{u}{c} = \alpha \beta_\parallel$$

Pulsation de Larmor non relativiste: $\Omega_0 = \dfrac{eB_0}{m_e}$

Condition de résonance: $\omega - \underline{k} \cdot \underline{v} = \Omega_0 / \gamma_0$

Soit, en projection: $\omega + k_\parallel u = \Omega_0 / \gamma_0$

Il est alors d'usage de définir $\quad x = \dfrac{\Omega_0/\gamma_0 - \omega}{k_\parallel u}$

Il y a résonance lorsque: $\begin{cases} x = -1 & \text{avec l'onde progressive} \\ x = +1 & \text{avec l'onde rétrograde.} \end{cases}$

Bien que dans la cavité, l'onde stationnaire doit une superposition de ces 2 ondes, c'est la condition x = -1 qui est intéressante en pratique. (Voir la discussion dans [1].)

En fonction de x, le champ magnétique s'écrit: $B_0 = \dfrac{m_e}{e} \gamma_0 (\omega + x k_\parallel u)$

Le courant d'accrochage en Ampères est alors donné par

$$I = \frac{\varepsilon_0 m_e}{e} \frac{\omega}{Q_{mpq}} |P_0|^2 \frac{(k_\parallel u)^2}{G(R_e)} \left[s \beta_\perp^2 \frac{dF_c}{dx} - 2 F_c \right]$$

avec:

$$|P_0|^2 = \frac{\pi L}{2} k^{-2} (\nu_{mp}^2 - m^2) \; J_m^2(\nu_{mp})$$

$$G(R_e) = J_{m\pm 1}^2 (k_\perp R_e)$$

$$F_c(x,q) = \frac{2}{(1-x^2)^2} \sin^2 \frac{(x+1)\pi q}{2}$$

$$s = \frac{\Omega_0}{\gamma_0 k_\parallel u}$$

Q_{mpq} = facteur de qualité du mode TE_{mpq}

$J_{m\pm 1}$: le signe dépend de la polarisation du mode ($\pm i m \phi$).
Donc pour $m \neq 0$, il y aura 2 courbes $I(B_0)$.

Allure de $F_c(x,q)$:



En posant $F_c = 2f^2$ avec $f = \frac{1}{1-x^2} \sin \frac{(x+1)\pi q}{2}$

On a $\lim\limits_{x \to -1} f = \frac{\frac{x+1}{2}\pi q}{(1+x)(1-x)} = \lim\limits_{x\to -1} \frac{\pi q}{2} \frac{1}{1-x} = \frac{\pi q}{4}$ d'où $\lim\limits_{x\to -1} F_c = \frac{\pi^2 q^2}{8}$

$F' = 4ff'$

$\lim\limits_{x\to -1} f' = \frac{\pi q}{2} \lim\limits_{x\to -1} \left(\frac{1}{1-x}\right)' = \lim\limits_{x\to -1} \frac{\pi q}{2} \frac{1}{(1-x)^2} = \frac{\pi q}{8}$

d'où $\lim F_c' = 4 \frac{\pi q}{4} \frac{\pi q}{8} = \frac{\pi^2 q^2}{8}$

Pour $x \to +1$: $\lim f = -\frac{\pi q}{4}$; $\lim F_c = \frac{\pi^2 q^2}{8}$; $\lim F' = \frac{-\pi^2 q^2}{8}$.

3. FICHIER DE DONNEES

Variable

NCAS	Nombre de cas à étudier
MAXQ	Valeur max. de l'index axial q
MM(1 à NCAS)	Index azymuthal (0, 1,...)
PP(1 à NCAS)	Index radial (1, 2,...)
LL(1 à NCAS)	Longueur d'intéraction [m] (1.5·FWHM)
QQ(1 à NCAS)	Facteur de qualité
V0	Tension d'accélération du faisceau [V]
ALPHA	Anisotropie
RE	Rayon du faisceau [m]
A	Rayon de la cavité [m]
BMIN, BMAX	Limites du champ magnétique [Tesla] (dessin)
IAMAX	Limite max du courant [A] (calcul et dessin)

Exemple:

```
M I T   1 4 0   G H Z   G Y R O T R O N
=======================================
 $GYRO
 NCAS = 3, MAXQ = 2,
 MM(1)=0, PP(1)=3, QQ(1)=1450., LL(1)=0.0150,
 MM(2)=2, PP(2)=3, QQ(2)=1400., LL(2)=0.0150,
 MM(3)=5, PP(3)=2, QQ(3)=1300., LL(3)=0.0150,
 V0=65000., ALPHA=1.5, RE=0.00182, A=0.00347,
 BMIN=5.2, BMAX=6.0, IAMAX=3.5,
 $END
```

4. Benchmark

Avec les données précédentes, on trouve:

			Expérimental [3]	
Mode	I	B	I	B
(0,3,1)	0.15	5.523	0.4	5.58
(2,3,1)	0.18	5.411		
(-2,3,1)	0.16	5.411	0.4	5.41
(+5,2,1)	0.48	5.714		
(-5,2,1)	0.11	5.714		
	[A]	[Tesla]	[A]	[Telsa]

L'accord quantitatif est nettement moins bon qu'avec le code ISTART3, mais cela donne une idée du spectre des modes axiaux excités dans un gyrotron.

---

[3] High Efficiency Operation of a 140 GHz Gyrotron, K.E. Kreisher, J.B. Schutkeker, R.J. Temkin, and W.J. Mulligan, Proc. 8th Int. Conf. on MM Waves, Miami, 1983.

M I T 1 4 0 G H Z G Y R O T R O N

STARTING CURRENT VERSUS STATIC MAGNETIC FIELD PROGRAM ISTART4

CRPP - 88/07/11. 14.31.25.
GYROTRON PROJECT



RESONATOR RADIUS: 3.47 MM
GAMMAO: 1.1272
ALPHA: 1.5000
BEAM VOLTAGE: 65.00 KV
ANNULAR BEAM RADIUS: 1.82 MM
HARMONIC # : 1

MODE TE	L [MM]	Q	FREQUENCY [GHZ]
(0, 3, 1)	15.0	1450.	140.244
(0, 3, 2)	15.0	363.	141.309
(0, 3, 3)	15.0	161.	143.064
(2, 3, 1)	15.0	1400.	137.447
(2, 3, 2)	15.0	350.	138.532
(2, 3, 3)	15.0	156.	140.323
(5, 2, 1)	15.0	1300.	144.996
(5, 2, 2)	15.0	325.	146.025
(5, 2, 3)	15.0	144.	147.725

4 2   G H Z   G Y R O T R O N

CRPP - 88/07/09. 10.45.06.
GYROTRON PROJECT

STARTING CURRENT VERSUS STATIC MAGNETIC FIELD   PROGRAM ISTART4



RESONATOR RADIUS: 8.00 MM
GAMMAO: 1.1566
ALPHA: 1.5000
BEAM VOLTAGE: 80.00 KV
ANNULAR BEAM RADIUS: 6.00 MM
HARMONIC # : 1

MODE TE	L [MM]	Q	FREQUENCY [GHZ]
(0, 2, 1)	40.0	618.	42.010
(0, 2, 2)	40.0	155.	42.508
(2, 2, 1)	40.0	556.	40.172
(2, 2, 2)	40.0	139.	40.693
(5, 1, 1)	40.0	503.	38.447
(5, 1, 2)	40.0	126.	38.991

CRPP - 88/07/11. 14.20.52.
GYROTRON PROJECT

8 GHZ GYROTRON

STARTING CURRENT VERSUS STATIC MAGNETIC FIELD   PROGRAM ISTART4



RESONATOR RADIUS: 22.80 MM
GAMMAO: 1.1566
ALPHA: 1.8900
BEAM VOLTAGE: 80.00 KV
ANNULAR BEAM RADIUS: 11.18 MM
HARMONIC # : 1

MODE TE	L [MM]	Q	FREQUENCY [GHZ]
(0, 1, 1)	164.0	227.	8.071
(0, 1, 2)	164.0	57.	8.224
(0, 1, 3)	164.0	25.	8.474
(0, 1, 4)	164.0	14.	8.813

```
      PROGRAM ISTART4(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C
C     COMPUTATION OF THE STARTING CURRENT OF A CONVENTIONAL GYROTRON
C     OPERATING NEAR THE FUNDAMENTAL OF THE LARMOR FREQUENCY
C
C     PLOT ON VERSATEC (CYBER)
C
C     - MODE TE(M,P,Q)   (SIMPLE CAVITY)
C     - ANALYTICAL APPROXIMATION (LINEAR REGIME)
C     - MKSA UNITS
C     - DOCUMENTATION: INT 145/88
C
C     ANDRE PERRENOUD      CRPP/EPFL    VERSION 1    09 JULY 1988
C---------------------------------------------------------------------
C     PARAMETERS
C     ----------
C     NMAX            MAX NUMBER OF MODES (M,P)
C     NPOINT          MAX NUMBER OF B FIELD POINTS
C     MMAX            MAX AZYMUTHAL INDEX      ( MMAX < 51 )
C     IPMAX           MAX RADIAL INDEX
C     IQMAX           MAX AXIAL INDEX
C
C     INPUT DATA
C     ----------
C     NCAS            NUMBER OF (M,P) MODES TO BE CONSIDERED
C     MAXQ            MAX AXIAL INDEX
C                     TOTAL NUMBER OF CCOMPUTED URVES WILL BE NCAS*MAXQ
C     MM(1...NCAS)    AZYMUTHAL INDEX
C     PP(1...NCAS)    RADIAL INDEX
C     QQ(1...NCAS)    QUALITY FACTOR OF MODE TE(M,P,1)
C     LL(1...NCAS)    CAVITY LENGTH OF MODE TE(M,P,1)
C     A               RADIUS OF RESONATOR (CYLINDRICAL CAVITY)
C     BMIN            LOWEST VALUE OF B FIELD  (FOR THE PLOT)
C     BMAX            HIGHEST VALUE            ( "   "   " )
C     IAMAX           HIGHEST VALUE OF CURRENT ( "   "   " )
C     VO              BEAM VOLTAGE
C     ALPHA           BEAM ANISOTROPY  (VPERP/VPARA)
C     RE              BEAM RADIUS
C---------------------------------------------------------------------
C
      PARAMETER (NMAX=10, NPOINT=300, MMAX=5, IPMAX=5, IQMAX=5)
      INTEGER P, PP
      REAL IAMAX, IA, MUO, L, LL, LAMBDA, KRE, KPERP, KPARA,
     *     KPU, NUMP, JMNUMP, JMPN
      DIMENSION BO(NPOINT), IA(NPOINT), BJZP(MMAX+1,IPMAX), BJ(50)
      CHARACTER DATE*10,TIME*10,DATE1*10,TIME1*10, LABEL*80
C
      COMMON /TITRE1/ LABEL(2)
      COMMON /TITRE2/ DATE1,TIME1
      COMMON /BEAM/   VO, ALPHA, RE, GAMMAO
      COMMON /CAVITY/ A, M, P, IQ, ISIG, MAXQ
      COMMON /MODE/   NCAS, MM(NMAX), PP(NMAX), QQ(NMAX), LL(NMAX),
     *                FREQ(NMAX,IQMAX)
      COMMON /PAGE/   XA4,YA4,DXL,DXR,DYU,DYD, CH, XO,YO,AXEX,AXEY
C
      NAMELIST /GYRO/ NCAS, MM, PP, QQ, LL, MAXQ, BMIN,BMAX, IAMAX,
     *                VO, ALPHA, RE, A
C---------------------------------------------------------------------
C
```

```
C  1.  CAS STANDARD (BENCHMARK)
C
       DATA LABEL(1)/'M I T   1 4 0   G H Z   G Y R O T R O N'/
       DATA LABEL(2)/'======================================='/
       DATA NCAS/1/, MAXQ/3/
       DATA MM(1), PP(1), QQ(1), LL(1)/0, 3, 1450., 0.0150/
       DATA MM(2), PP(2), QQ(2), LL(2)/2, 3, 1400., 0.0150/
       DATA MM(3), PP(3), QQ(3), LL(3)/5, 2, 1300., 0.0150/
       DATA VO, ALPHA, RE, A/ 65000., 1.5, 1.82E-03, 3.47E-03/
       DATA BMIN, BMAX, IAMAX/ 5.2, 6.0, 7.0/
C
       DO 10 NC= NCAS+1, NMAX
       MM(NC)=0
       PP(NC)=0
       QQ(NC)=0.
10     LL(NC)=0.
C------------------------------------------------------------------
C
C  2.  CONSTANTS
C
       DATA CLIGHT/2.99792458E+08/
       DATA EMASS,ECHARG/9.1093897E-31, 1.60217733E-19/
       PI   = ACOS(-1.)
       MUO  = 4.E-07*PI
       EPSO = 1./(MUO*CLIGHT**2)
C
       DATE1=DATE()
       TIME1=TIME()
       WRITE(6,100) DATE1,TIME1
100    FORMAT('1'////' PROGRAM ISTART4',4X,2A10/
     *               ' ===============''//)
C
       WRITE (6,105)
105    FORMAT (' ZEROES OF DERIVATIVES OF BESSEL FUNCTIONS'/)
       CALL BEPZERO(BJZP,MMAX,IPMAX)
C------------------------------------------------------------------
C
C  3.  READ INPUT DATA
C
       REWIND (5)
       READ (5,'(A)') LABEL
       WRITE (6,'(//1X,A/1X,A/)') LABEL
       READ (5,GYRO)
       WRITE (6,GYRO)
C
       GAMMAO = 1. + (ECHARG*VO)/(EMASS*CLIGHT**2)
       BPARA  = SQRT ((1.-1./GAMMAO**2)/(1.+ALPHA**2))
       BPERP  = ALPHA * BPARA
       WRITE (6,110) 'GAMMAO =',GAMMAO
110    FORMAT (1X,A,G14.6)
       WRITE (6,110) 'BETA PERP =', BPERP
       WRITE (6,110) 'BETA PARA =', BPARA
C
       CF = EPSO*EMASS/ECHARG
       WRITE (6,110) 'CONST =', CF
C------------------------------------------------------------------
C
C  4.   CALCULS
C
       IF (MAXQ.GT.IQMAX) STOP 'AXIAL INDEX Q TOO LARGE'
       IFIRST = 1
```

```
      DO 490 NC=1,NCAS
      M = MM(NC)
      P = PP(NC)
      Q = QQ(NC)
      L = LL(NC)
      IF (M .GT. MMAX)  STOP ' M TOO LARGE'
      IF (P .GT. IPMAX) STOP ' P TOO LARGE'
      NUMP = BJZP(M+1,P)
      CALL BESSEL (M, NUMP,BJ)
      JMNUMP = BJ(M+1)
C
      WRITE (6,120) M,P
120   FORMAT (///' MODES TE(',I2,',',I2,',*)'/
     *        ' ================'/)
      WRITE (6,110) ' INTERACTION LENGTH=', L
      WRITE (6,110) ' NUMP   =    ', NUMP
      WRITE (6,110) ' JM (NUMP) = ', JMNUMP
C
      KPERP = NUMP/A
      WRITE (6,110) ' KPERP      ', KPERP
      KRE   = KPERP*RE
      WRITE (6,110) ' KPERP*RE   ', KRE
      PO2 = 0.5*PI*L*(NUMP**2 - M**2)*JMNUMP**2 /KPERP**2
C
      NSIG = 2
      IF (M.EQ.0) NSIG=1
      DO 480 NS=1,NSIG
      ISIG = 3 - 2*NS
      MN = M + ISIG
      IF (M.NE.0 ) WRITE (6,435) ISIG
435   FORMAT (/'  ROTATING MODE: ',I2)
      CALL BESSEL(MN,KRE,BJ)
      JMPN = BJ(MN+1)
      WRITE (6,110) ' JMPN  =    ', JMPN
      GRE = JMPN**2
      WRITE (6,110) ' G(RE) =    ', GRE
C
      DO 470 IQ= 1,MAXQ
      WRITE (6,125) M,P,IQ
125   FORMAT (///' MODE TE(',I2,',',I2,',',I2,')'/
     *        ' ----------------'/)
      KPARA = (IQ*PI)/L
      WRITE (6,110) ' KPARA      ', KPARA
      OMEGA = CLIGHT * SQRT(KPERP**2 + KPARA**2)
      QMPQ = Q/(IQ**2)
      FREQ(NC,IQ) = OMEGA/(2.*PI)
      WRITE (6,110) ' FREQUENCY   =    ', FREQ(NC,IQ)
      LAMBDA = CLIGHT/FREQ(NC,IQ)
      WRITE (6,110) ' WAVELENGTH =    ', LAMBDA
      WRITE (6,110) ' QUALITY FACTOR =    ', QMPQ
      KPU = KPARA*BPARA*CLIGHT
      WRITE (6,110) ' KPU        ', KPU
C-----------------------------------------------------------------------
C
C  5.  COMPUTES POINTS (BO, IA) FOR -3. < X < 0.
C
      J = 0
      DX = 3./NPOINT
      X = -3.
430   X = X+ DX
      IF (X.GE.0.) GOTO 440
```

```
      J = J+1
      IF (J.GE.NPOINT) GOTO 440
      OMEGAO = GAMMAO *(OMEGA + X*KPU)
      BO(J) = EMASS/ECHARG *OMEGAO
      S = OMEGAO / (KPU*GAMMAO)
      CALL CFCQ(X,IQ,PI, FC,DFC)
      IA(J) = CF*OMEGA/QMPQ *PO2 *KPU**2/GRE /(S*BPERP**2*DFC -2.*FC)
      GOTO 430
440   WRITE (6,445)
445   FORMAT (/10X,'BO [TESLA]   IA START [A]'/)
      WRITE (6,450) (J1, BO(J1), IA(J1), J1=1,J)
450   FORMAT (I5,2E15.4)
C-----------------------------------------------------------------------
C
C   6. PLOT ON VERSATEC
C
C     IF (IA(1).GT.IAMAX .AND. IA(2).LT.IAMAX) THEN
C       BO(1) = BO(2) + (IAMAX-IA(2))*(BO(1)-BO(2))/(IA(2)-IA(1))
C       IA(1) = IAMAX
C     ENDIF
C     IF (IA(J).GT.IAMAX .AND. IA(J-1).LT.IAMAX) THEN
C       BO(J) = BO(J-1) - (IAMAX-IA(J-1))*(BO(J-1)-BO(J))/
C    *                    (IA(J)-IA(J-1))
C       IA(J) = IAMAX
C     ENDIF
      CALL PLOTST(BO,IA,J, BMIN,BMAX,IAMAX,IFIRST)
C
470   CONTINUE
480   CONTINUE
490   CONTINUE
      CALL TITLE
      CALL PLOT(0.,0.,999)
      STOP
      END
```

```
      SUBROUTINE CFCQ(X,IQ,PI, FC,DFC)
C
      PIQ2 = 0.5*PI*IQ
      UNPX = 1.+X
      UNMX = 1.-X
      IF (ABS(UNPX).LT.0.001) THEN
        F  = 0.5*PIQ2
        DF = 0.5*F
      ELSE IF (ABS(UNMX).LT.0.001) THEN
        F  = 0.5*PIQ2
        DF =-0.5*F
      ELSE
        UNMX2 = UNPX*UNMX
        F = SIN(PIQ2*UNPX) / UNMX2
        DF = (PIQ2*COS(PIQ2*UNPX) + 2.*X*F) / UNMX2
      ENDIF
C
      FC  = 2.*F*F
      DFC = 4.*F*DF
      END
```

```
      SUBROUTINE BEPZERO(BJZP,M,N)
C
C     COMPUTES THE ZEROES OF THE DERIVATIVES OF BESSEL FUNCTION
C     OF INTEGER ORDER (M=0,1,2...5)
C
C     ON RETURN:  BJZP(K+1, L) = LTH ZERO OF J'K(X)    (L = 1,2 3...N)
C
      PARAMETER (EPS=1.E-12,  MMAX=5,  IPMAX=5)
C
      DIMENSION BJZP(MMAX+1,IPMAX),  BJ(50)
C
      PI =ACOS(-1.)
      IF (M.GE.0 .AND.  M.LE.MMAX) GOTO 5
      WRITE (6,1) M
1     FORMAT (' SUBROUTINE BEPZERO CALLED WITH M=',I3)
      STOP
5     IF (N.GE.1 .AND.  N.LE.IPMAX) GOTO 10
      WRITE (6,2) N
2     FORMAT (' SUBROUTINE BEPZERO CALLED WITH N=',I3)
      STOP
C
C     FIND ROOTS USING NEWTON ALGORITHM FOR J'M(X)=0
C
10    DO 30 IM=0,M
      DO 30 IN=N,1,-1
C
      IF (IN.GE.N-1) THEN
C                         USE ASYMPTOTIC ROOTS AS STARTING VALUES
        XO = PI/4. * (2*IM + 4*IN -3)
        IF (IM.EQ.0) XO = XO + PI
      ELSE
        XO = 2*BJZP(IM+1,IN+1) - BJZP(IM+1,IN+2)
      ENDIF
      ITER=0
C
C     NEWTON SCHEME:  X1 = XO - JM'(XO)/JM"(XO)
C
15    CALL BESSEL(IM+2,XO,BJ)
      BJM    = BJ(IM+1)
      BJMP   = -BJ(IM+2) + IM/XO*BJM
      BJMP1P = -BJ(IM+3) + (IM+1)/XO*BJ(IM+2)
      BJM2P  = -BJMP1P + IM/XO*(BJMP - BJM/XO)
      X1 = XO - BJMP/BJM2P
      ITER = ITER +1
      IF (ABS(X1-XO) .LT. EPS) GOTO 20
      XO=X1
      IF (ITER.LT.50) GOTO 15
      WRITE (6,16) IM,IN
16    FORMAT (' NO CONVERGENCE REACHED FOR M=',I3,'  N=',I3)
      X1=0.
20    BJZP(IM+1,IN)=X1
      WRITE (6,17) IM,IN,X1,ITER
17    FORMAT (' J''(',I3,',',I3,') =',F12.8,'  ITER=',I3)
30    CONTINUE
      END
```

```
      SUBROUTINE BESSEL(LMAX,X,BJ)
C
C     CALCULATE BESSEL FUNCTIONS BY BACKWARD RECURSION
C
C     ON RETURN:  BJ(1)=J0(X),  BJ(2)=J1(X),  ...  BJ(LMAX+1)=BJLMAX(X)
C
C
      PARAMETER (XMIN=.00001,XMAX=100.,MSTART=200,DEL=1.E-60)
      DIMENSION TEMP(MSTART),BJ(*)
C
      IF ( LMAX .GT.50  ) STOP 'BESSEL: ORDER TOO BIG'
      IF (ABS(X).GT.XMAX) STOP 'BESSEL: ARGUMENT TOO BIG'
      IF (ABS(X).GT.XMIN) GOTO 20
      BJ(1)=1.
      IF (LMAX.EQ.0) RETURN
      DO 10 L=1,LMAX
10    BJ(L+1)=X/FLOAT(2*L)*BJ(L)
      RETURN
C
20    LSTART=2*(5*INT(ABS(X))+(LMAX+1)/2+10)+1
C.....ORDER OF HIGHEST BESSEL FUNCTION IS EVEN TO GUARANTEE CORRECT
C.....PARITY (JN(X) = (-1)**N *JN(-X) )
      IF (LSTART.GT.MSTART) LSTART=MSTART
      TEMP(LSTART)=DEL
      TEMP(LSTART-1)=DEL*FLOAT(2*(LSTART-2))/X
      TS=TEMP(LSTART)**2
      DO 30 L=LSTART-1,2,-1
      TEMP(L-1)=FLOAT(2*(L-1))/X*TEMP(L)-TEMP(L+1)
30    TS=TS+TEMP(L)**2
      TS=1./SQRT(TEMP(1)**2+2.*TS)
      DO 40 L=1,LMAX+1
40    BJ(L)=TS*TEMP(L)
      RETURN
      END
```

```
      SUBROUTINE PLOTST(BO,IA,J, BMIN,BMAX,IAMAX,IFIRST)
C
      COMMON /PAGE/   XA4,YA4,DXL,DXR,DYU,DYD, CH, XO,YO,AXEX,AXEY
C
      REAL BO(*), IA(*), IAMAX, IAKM1
      INTEGER P
      CHARACTER T*20
C
      COMMON /CAVITY/ A, M, P, IQ, ISIG, MAXQ
C
      DATA XA4,YA4/29.,21./, DXL,DXR,DYU,DYD/3.,2.,2.5,3./, CH/0.2/
      DATA XO,YO,AXEX,AXEY/3.,2.5,16.,14./
C
      IF (IFIRST.EQ.0) GOTO 25
      CALL PLOTS ('VERSATEC',100.,8224,0,1)
C
      CALL FRAME(BMIN,BMAX,0.,IAMAX, XO,YO,AXEX,AXEY)
      CALL CHPEN(3)
      IFIRST=0
C
25    CONTINUE
      IPEN=3
      IPL = 0
      DO 30 K=1,J
      CURR=IA(K)
      IF (CURR.LT.0. .OR. CURR.GT.IAMAX) THEN
        IPEN=3
        IPL =0
        IAKM1=IAMAX+1.
      ELSE
        XX = XO + (BO(K)-BMIN)/(BMAX-BMIN) * AXEX
        YY = YO + CURR/IAMAX * AXEY
        CALL PLOT(XX,YY,IPEN)
        DIA = CURR-IAKM1
        IAKM1=CURR
        IF (DIA.GT.0. .AND. IPL.EQ.0 .AND. IPEN.EQ.2) THEN
C         WRITE MODE #
          XXP = XX - 1.2
          YYP = YY - 0.5
          IF (YYP.LT.(YO+0.5)) YYP = YYP + 1.
          WRITE (T,50) ISIG*M, P, IQ
50        FORMAT('(',I2,',',I2,',',I2,')')
          CALL SYMBOL(XXP,YYP,0.2,T,0.,10)
          IPL = 1
          CALL PLOT(XX,YY,3)
        ENDIF
        IPEN=2
      ENDIF
30    CONTINUE
      END
```

DO=-LONG/-OT, ARG=-COMMON/-FIXED, CS= USER/-FIXED, DB=-TB/-SB/-SL/-ER/-ID/-PMD/-ST, -AL, PL=5000
FTN5, I=ISTART4, L=LIS, ANSI=0, LO=S/-A, PN.

```
      SUBROUTINE FRAME(BMIN,BMAX,ISMIN,ISMAX)
      COMMON /PAGE/   XA4,YA4,DXL,DXR,DYU,DYD,  CH,  XO,YO,AXEX,AXEY
C
      REAL ISMIN, ISMAX
C
      DATA MASK/O"52525252525252525252"/
C
C     CH = CHARACTER HEIGHT
C
C     A4 FRAME
C
      CALL CHPEN(2)
      CALL PLOT(0.    ,1.     ,3)
      CALL PLOT(0.    ,0.     ,2)
      CALL PLOT(1.    ,0.     ,2)
      CALL PLOT(XA4-1.,0.     ,3)
      CALL PLOT(XA4   ,0.     ,2)
      CALL PLOT(XA4   ,1.     ,2)
      CALL PLOT(XA4-1.,YA4    ,3)
      CALL PLOT(XA4   ,YA4    ,2)
      CALL PLOT(XA4   ,YA4-1. ,2)
      CALL PLOT(0.    ,YA4-1. ,3)
      CALL PLOT(0.    ,YA4    ,2)
      CALL PLOT(1.    ,YA4    ,2)
C
C     AXES AND GRID
C
      DB=(BMAX-BMIN)/AXEX
      CALL AXIS(XO,YO,AXEX,0.,BMIN,DB,'MAGNETIC FIELD (TESLA)',22)
C
      DIS=(ISMAX-ISMIN)/AXEY
      CALL AXIS(XO,YO,AXEY,90.,ISMIN,DIS,'STARTING CURRENT (A)',-20)
      CALL CHPEN(3)
      CALL PLOT(XO,YO,3)
      CALL PLOT(XO,YO+AXEY,2)
      CALL PLOT(XO+AXEX,YO+AXEY,2)
      CALL PLOT(XO+AXEX,YO,2)
      CALL PLOT(XO,YO,2)
      CALL CHPEN(1)
      IAXEX=AXEX
      IAXEY=AXEY
      CALL GRID(XO,YO,IAXEX,1.,IAXEY,1.,MASK)
      CALL CHPEN(2)
      RETURN
      END
```

```
      SUBROUTINE TITLE
C
      PARAMETER (NMAX=10, IQMAX=5)
      COMMON /PAGE/   XA4,YA4,DXL,DXR,DYU,DYD, CH, XO,YO,AXEX,AXEY
      COMMON /TITRE1/ LABEL(2)
      COMMON /TITRE2/ DATE1,TIME1
      COMMON /BEAM/   VO, ALPHA, RE, GAMMAO
      COMMON /CAVITY/ A, M, P, IQ, ISIG, MAXQ
      COMMON /MODE/   NCAS, MM(NMAX), PP(NMAX), QQ(NMAX), LL(NMAX),
     *                FREQ(NMAX,IQMAX)
C
      INTEGER P, PP
      REAL LL
      CHARACTER T*80,DATE1*10,TIME1*10,LABEL*80
      YTOP=YA4-DYU
      CALL CHPEN(2)
      CALL SYMBOL(DXL,YTOP,CH,LABEL(1),0.,80)
      CALL SYMBOL(DXL,YTOP-2.*CH,CH,LABEL(2),0.,80)
      WRITE (T,100)
100   FORMAT ('STARTING CURRENT VERSUS STATIC MAGNETIC FIELD',5X,
     *        'PROGRAM ISTART4    ')
      CALL SYMBOL(DXL,YTOP-5.*CH,CH,T,0.,66)
C
      WRITE (T,1) DATE1,TIME1
1     FORMAT ('CRPP - ',2A10)
      CALL SYMBOL(0.8*XA4,YTOP,0.8*CH,T,0.,27)
      WRITE (T,2)
2     FORMAT ('GYROTRON PROJECT')
      CALL SYMBOL(0.8*XA4,YTOP-2.*CH,0.8*CH,T,0.,16)
C
      XT=XO+AXEX+1.
      YT=YO+AXEY
      WRITE (T,3) A*1000.
3     FORMAT('RESONATOR RADIUS:',F7.2,' MM')
      CALL SYMBOL(XT,YT,CH,T,0.,27)
      YT=YT-2.*CH
      WRITE (T,7) GAMMAO
7     FORMAT('GAMMAO: ',F7.4)
      CALL SYMBOL(XT,YT,CH,T,0.,14)
      YT=YT-2.*CH
      WRITE (T,8) ALPHA
8     FORMAT('ALPHA:  ',F7.4)
      CALL SYMBOL(XT,YT,CH,T,0.,14)
      YT=YT-2.*CH
      WRITE (T,9) VO/1000.
9     FORMAT('BEAM VOLTAGE:',F6.2,' KV')
      CALL SYMBOL(XT,YT,CH,T,0.,22)
      YT=YT-2.*CH
      WRITE (T,10) RE*1000.
10    FORMAT('ANNULAR BEAM RADIUS:',F6.2,' MM')
      CALL SYMBOL(XT,YT,CH,T,0.,29)
      YT=YT-2.*CH
      WRITE (T,11) 1
11    FORMAT('HARMONIC # :',I2)
      CALL SYMBOL(XT,YT,CH,T,0.,14)
C
      YT=YT-3.*CH
      T='   MODE          L       Q   FREQUENCY'
      CALL SYMBOL(XT,YT,CH,T,0.,35)
```

```
      YT=YT-2.*CH
      T='    TE          [MM]             [GHZ]   '
      CALL SYMBOL(XT,YT,CH,T,0.,35)
      DO 20 NC=1,NCAS
      DO 20 IQQ=1,MAXQ
      YT=YT-2.*CH
      WRITE (T,21) MM(NC), PP(NC), IQQ, LL(NC)*1000., QQ(NC)/IQQ**2,
     *             FREQ(NC,IQQ)/1.E+09
21    FORMAT('(',I2,',',I2,',',I2,') ',F7.1,F7.0,F9.3)
20    CALL SYMBOL(XT,YT,CH,T,0.,34)
C
      RETURN
      END
```